

## Paper 043-2007

## Using Arrays to Calculate Medication Utilization

R. Scott Leslie, MedImpact Healthcare Systems, Inc., San Diego, CA

**ABSTRACT**

Assessing duration of medication therapy involves managing a data set with multiple observations per subject. This paper offers an innovative approach to calculating medication utilization as the proportion of days supplied over a specified time period. In this paper, the TRANSPOSE procedure, ARRAY statements, and DO loops are used to create multiple indicator variables, which are then used to calculate medication utilization. Variations of this code can integrate gaps and overlaps in therapy and can be used in calculating concomitant medication utilization.

**INTRODUCTION**


Many health outcomes related to pharmacy utilization involve length of therapy measurements. The purpose of this paper is to offer code that calculates a patient's medication utilization as the proportion of days medication is supplied over a time period. This code is a helpful start for building code to calculate additional outcome measures such as compliance, adherence, and persistence.

**EXAMPLE 1: PROPORTION OF DAYS MEDICATION SUPPLIED**

This example uses a pharmacy claims data set that has multiple observations per patient. The steps below calculate the number of days a single drug is supplied over a 180-day study period, with the date of first claim as the first day of study period. A cut of the data set shows 3 claims for a patient.

<u>Obs</u>	<u>member_id</u>	<u>fill_dt</u>	<u>drug</u>	<u>days_supply</u>
946	603	02/17/2005	a	30
947	603	06/13/2005	a	30
948	603	08/11/2005	a	30

Figure 1 illustrates the above data showing 3 claims and the 180 day study period. Date of first fill is the start of study period and end of study period is 180 days post first fill.

**STEP 1**

The first step is to transpose the data to a single observation per patient data set. This is done twice for the purposes of detailing the fill dates and corresponding days supply for each fill. It is essential to sort the data set by patient and fill date. Start and end dates for each subject are also calculated.

```
proc sort data=claims;
  by member_id fill_dt;
```

```

run;

proc transpose data = claims out=fill_dates (drop=_name_) prefix = fill_dt;
by member_id;
var fill_dt;
run;

proc transpose data = claims out=days_supply (drop=_name_) prefix = days_supply;
by member_id;
var days_supply;
run;

data both;
merge fill_dates days_supply;
by member_id;
format start_dt end_dt mmddyy10.;
start_dt=fill_dt1;
end_dt=fill_dt1+179;
run;

```

The result of the above code creates a patient level data set, showing the medication fill pattern and days supply for each fill. Note that missing values are given to those variables where the variable being transposed has no value in the input data set. That is, this patient has three claims; therefore the values for fill\_dt3 and fill\_dt4 are missing.

Obs	member_id	fill_dt1	fill_dt2	fill_dt3	fill_dt4	fill_dt5
265	603	02/17/2005	06/13/2005	08/11/2005	.	.

Obs	days_supply1	days_supply2	days_supply3	days_supply4	days_supply5
265	30	30	30	.	.

Obs	start_dt	end_dt
265	02/17/2005	08/15/2005

## STEP 2

Next, a data step uses arrays and DO loops to find the days the patient was supplied the medication and calculates the proportion of days the medication was supplied in the review period. The first array, daydummy, creates a dummy variable for each day in the review period. The next two arrays, groups the fill\_dt and days\_supply variables setting up the DO loops. In this data set, the maximum number of fills incurred by a patient was 11 so there are 11 elements for these two arrays. One can set the number of elements to a value beyond the reasonable amount of fills or get the maximum number of fills in the data set from a proc contents procedure. The first do loop sets each dummy variable, daydummy, to 0. The second do loop uses an IF statement to flag the days of the review period that the patient was supplied the medication. Next, the variable dayscovered sums the daydummy variables. This sum is used as the numerator in calculating, p\_dayscovered, the proportion of days medication was supplied in the 180 day study period, which is one of many measures for assessing compliance.

```

data pdc;
set both;
array daydummy(180) day1-day180;
array filldates(*) fill_dt1 - fill_dt11;
array days_supply(*) days_supply1-days_supply11;

do ii=1 to 180; daydummy(ii)=0;end;

do ii=1 to 180;
do i = 1 to dim(filldates) while (filldates(i) ne .);
if filldates(i)<= start_dt + ii -1 <= filldates(i)+days_supply(i)-1
then daydummy(ii)=1;
end;
end;
drop i ii;
dayscovered=sum(of day1 - day180);label dayscovered='Total Days Covered';
p_dayscovered=dayscovered/180;label p_dayscovered='Proportion of Days Covered';
run;

```


```
proc print data=pcd;where member_id=603;run;
```

The result is a data set that has 180 dummy variables, one for each day of the time period, which indicates medication supplied. Only a few of the dummy variables are displayed below. In this example the patient's last fill\_date is on day 176 (see Figure 1 above) with a majority of the days supply for this claim extending beyond the study period. The claim is truncated and only 5 of days of this claim are included in the days covered count.

Obs	member_id	day1	day2	day3	day4	day5	***day6-day29***	day30	day31	day32	day33
265	603	1	1	1	1	1	1	1	0	0	0
Obs	***day34-day115***	day116	day117	day118	day119	***day120-day145***	day146				
265	0	0	1	1	1	1	1				
Obs	***day147-day174***	day175	day176	day177	day178	day179	day180				
265	0	0	1	1	1	1	1				
Obs	dayscovered	p_dayscovered									
265	65	0.36111									

### EXAMPLE 2: CREDITING OVERLAPPING DAYS SUPPLY

In this example, a patient refills their medication before exhausting the previous fill. Figure 2 shows this scenario, where the fourth claim, filled on 7/30/05, occurs before the end of supply of the third claim (8/5/05). Proportion of days supplied is calculated in the same manner and credits the subject with finishing the previous fill. This code is similar to the previous example with one extra step that identifies the overlapping days supply and shifts the fill date forward to the day after the end of supply of the previous fill.



### ADDITIONAL STEP


Using the same steps as in Example 1, adding this additional DO loop adjusts fill dates by shifting them forward. This starts with the second fill:

```
do u=2 to 11 while (filldates(u) ne .);
  if filldates(u)<filldates(u-1)+days_supply(u-1)
  then filldates(u)=filldates(u-1)+days_supply(u-1);
end;
```

Shifting the fourth fill date credits the patient with 7 more days supply, increasing the days covered from 113 to 120 and increasing proportion of days supplied from to 62.8% to 66.6%. Larger differences would be seen in cases where a patient has multiple claims and multiple overlaps.

### EXAMPLE 3: CONCOMITANT THERAPY

With a few additional steps this code can calculate simultaneous coverage of multiple medications over a specified time period. Figure 3 shows a scenario where a patient is using two medications for the majority of a 180-day period; however, synchronized coverage of the medications occurs in the less than half of the days.



#### ADDITIONAL STEP

To assess simultaneous coverage of two medications, the code in Example 1 can be replicated for the second medication. Two data sets, each with a set of day dummy variables, would be created. Merging the data sets and the use a DO loop would then sum the days that both medications were covered.

```
data pdc_1_2;
merge pdc_1 pdc_2;
by member_id;
array drug1 (180) day1 - day180;
array drug2 (180) drug2day1 - drug2day180;
do i = 2 to 180;
    if drug1(i) = 1 AND drug2(i) = 1 then dayscovered_both = dayscovered_both + 1;
end;
label dayscovered_both = 'Total Days Covered on Both Drugs';
p_dayscovered_both= dayscovered_both / 180;
label p_dayscovered_both = 'Proportion of Days Covered on Both Drugs';
run;
```

In this example, coverage on Therapy A alone is 120 days/180 days, or 67%. Coverage on Therapy B alone is 120 days/180 days, or 67%. However, simultaneous coverage is only 72 days/180 days or 40%.

### CONCLUSION

This paper offers examples of calculating duration of therapy, as proportion of days supplied over a time period. Using dummy variables to specify treatment for each day of a review period, you can identify various outcomes such as compliance, adherence, persistence, and discontinuation to therapy.

### REFERENCES

SAS Institute Inc. 2004. "Language Reference Concepts: Array Processing". *SAS OnlineDoc*® 9.1.3. Cary, NC: SAS Institute Inc.  
[http://support.sas.com/documentation/onlinedoc/91pdf/sasdoc\\_913/base\\_lrconcept\\_9196.pdf](http://support.sas.com/documentation/onlinedoc/91pdf/sasdoc_913/base_lrconcept_9196.pdf) (April 2, 2006)

SAS Institute Inc. 2004. "SAS Procedures: The TRANSPOSE Procedure". *SAS OnlineDoc*® 9.1.3. Cary, NC: SAS Institute Inc.

[http://support.sas.com/documentation/onlinedoc/91pdf/sasdoc\\_913/base\\_proc\\_8977\\_new.pdf](http://support.sas.com/documentation/onlinedoc/91pdf/sasdoc_913/base_proc_8977_new.pdf)

### **ACKNOWLEDGMENTS**

The author would like to thank Patrick Thiebaud for explaining the concept of medication coverage and for his programming advice.

### **CONTACT INFORMATION**

Your comments and questions are valued and encouraged. Contact the author at:

Scott Leslie  
MedImpact Healthcare Systems, Inc.  
10680 Treena Street  
San Diego, CA 92131  
Work Phone: 858-790-6685  
Fax: 858-689-1799  
Email: [scott.leslie@medimpact.com](mailto:scott.leslie@medimpact.com)

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.